
1 Giriş

C-Gate ’un bazı işlevleri için Carola programından C-Gate ’a bilgi verilmesi gerekir. İlgili işlevsellik için gerekli bloklar ve
talimatlar için lütfen ilgili bölümlere bakın.

ReadVar veri yapısını veya OPCUAVAR komutunu başka bir bağlamda kullanırsanız, kullanılan alanların çakışmadığın-
dan emin olun. Ayrıca Carola programında kullanılan indeksin web arayüzündeki indeksle aynı olmadığını lütfen unutma-
yın. Değerler C-Gate adresine ReadVar veri yapısı üzerinden aktarılır. Kontrolörde ilk indeks bir ve son 100’dür. C-Gate
’da ilk indeks 0 ve son 99’dur. Kontrolörde 50 indeksi altında saklanan bir değer C-Gate ’da 49 indeksi altında alınabilir.

Lütfen aşağıda açıklanan Carola kod örneklerinin kısmen veya tamamen Yüksek Lisans programına kaydedilmiş olabilece-
ğini unutmayın. Lütfen bölümleri karşılaştırın ve gerekirse uyarlayın. Fonksiyonların ve değişkenlerin tekrarlanması Carola
yorumlayıcısı tarafından bir hata olarak algılanır.

Not: QIROX WriteVar değişkenlerinde yapılan değişiklikler Carola program sırasını etkileyebilir.

Not: Carola program kodunu kopyalamak istiyorsanız lütfen bu dosyayı açmak için Adobe Acrobat Reader kullanın.

2 Genel Ekipman Etkinliği (OEE)

2.1 Gerçekmiktarı ve iyi miktarı belirleyin

Aşağıdaki program bölümleri, gerçek miktarı ve iyi miktarı belirlemek için sırasıyla ana programa ve bileşen programına
kaydedilmelidir. Aşağıdaki prosedürlerin sadece 94 ila 100 endeks aralığını kullanması önemlidir ve başka amaçlar için kul-
lanılamaz. Bu aralık gelecekte genişleyebilir.

2.1.1 Yüksek Lisans Programı

İlk satırlar, değişkenlerin diğer programlardan alınabilmesi ve yazılabilmesi için gereklidir. Bunlar RESTART komutundan
hemen sonra yer alır.

RESTART
! ĞDEISKEN BILDIRIMI
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM

Bu blok, C-Gate adresine bir bileşenin başladığına dair bilgi gönderen bir prosedür tanımlar. Bundan önce, aşağıda kullanı-
lan değişkenler bileşen programında ayarlanmalıdır.

PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'

STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('ĞDEISKEN >COMPTYPE< HENUEZ TANIMLANMAMIS!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
ENDP

Bu blok, bir bileşenin tamamlandığı bilgisini C-Gate adresine ileten bir prosedür tanımlar.

! BILESEN KAYDININ SONU
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)

ENDP

Bublok,C-Gateadresineyenibir dikişinbaşladığı bilgisini aktaranbir fonksiyon tanımlar. Prosedür çağrılmadanöncedikişin
adı COMPSEAM değişkenine yazılmalıdır.

! DIKISE BASLAMA
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Bu blok, C-Gate adresine mevcut dikişin bittiği bilgisini ileten bir fonksiyon tanımlar.

! BIR DIKISIN SONU
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Bu blok PHG üzerinde bileşenin uygun olup olmadığını sorgulayan bir fonksiyon tanımlar. Kontrol, çalışan tarafından yapı-
langörselbir incelemeyedayanır vebir giriş yapılanakadardaha fazlaprogramakışını engeller. BILESEN OK=1 MI YOKSA NOT OK=0 MI?
ifadesini gereksinimlerinize göre uyarlayabilirsiniz.

! KALITELI BILESEN KAYDI
PUBLIC PROC COMPQMAN
WREAD ('BILESEN OK=1 MI YOKSA NOT OK=0 MI?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

Bu blok, bileşenin kalitesini otomatik olarak (sorgu olmadan) Tamam olarak ayarlayan bir işlev tanımlar. Bu işlev, genel
ekipman etkinliğinin (OEE) kalitesini dinamik olarak kaydetmek istemiyorsanız, ancak yine de tüm metriğin doğru hesap-
lanmasını istiyorsanız kullanılabilir.

! KALITELI BILESEN KAYDI
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

2.1.2 Değişkenler

Programda aşağıdaki değişkenler kullanılabilir.

• COMPTYPE : Bileşen tipi.

• COMPVARI : Bileşenin varyantı.

• COMPSERI Bileşenin seri numarası. Herhangi bir değer ayarlanmazsa, C-Gate zaman damgasına dayalı olarak oto-
matik olarak bir seri numarası oluşturur.

• COMPSTEP : Üretim adımının adı. Bileşen birkaç istasyonda işleniyorsa değer önemlidir.

• COMPSEAM : Dikişin adı. Bir dikiş birkaç ağ segmentinden oluşabilir.

2.1.3 Basit bileşen programı

Lütfen değişkenleri ve prosedürleri ana programdan içe aktarın.

RESTART

! ĞDEISKENLERIN VE PROSEDUERLERIN IÇE AKTARILMASI
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER

MAIN

Program akışında, en azından zorunlu alanları ve gerekirse isteğe bağlı alanları da ayarlayın. Tüm alanlar ayarlandıktan
sonra COMPSTART prosedürünü çağırın. Eğer C-Gate.QM modülünü satın aldıysanız, ek olarak dikişleri tanımlayabilir ve
bileşen kaydını daha hassas bir şekilde yapılandırabilirsiniz. Bunu yapmak için SEAMSTRT ve SEAMEND prosedürlerini
kullanın. Parça tamamlandığında ve Carola programından parça kalitesini ayarlamak istediğinizde, kalite verilerini aktar-
mak için COMPQAUT veya COMPQMAN adreslerini arayın. Eğer C-Gate.QMmodülünü satın aldıysanız, parça kalitesi QIROX
SD izleme yardımıyla otomatik olarak belirlenir. COMPQAUT veya COMPQMAN prosedürünü çağırmanın parça protokolü
üzerinde hiçbir etkisi yoktur.

Bileşeni bitirmek için COMPEND prosedürünü çağırın. Çağrıların sırası önemlidir.

Lütfen COMPEND adresinin çağrılması ve bir bileşenin COMPSTART ile yeniden başlatılması arasında bir saniyeye kadar
süre geçmesi gerektiğini unutmayın, aksi takdirde kontrolördeki veri senkronizasyonu nedeniyle bitiş sinyali kaybolabilir.
Bu sınırlamanın robot kontrolörünün gelecek yazılım sürümlerinden birinde kaldırılması planlanmaktadır.

! BILESEN BILGILERININ AYARLANMASI
! ZORUNLU ALAN
COMPTYPE := 'KONTROL KABINI'

! OPSIYONEL
! COMPVARI := 'XL'

! OPSIYONEL
! COMPSERI := '1234567890'

! OPSIYONEL
! COMPSTEP := 'ZEMIN'

! BILESENIN BASLANGICI
CALL COMPSTART

! ***************
! BILESEN PROGRAMI
! ***************

COMPSEAM := 'DOLGU ĞKAYNAI 1'
CALL SEAMSTRT

! ***************
! PROGRAM DOLGU ĞKAYNAI 1. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.
! ***************

COMPSEAM := 'DOLGU ĞKAYNAI 2'
CALL SEAMSTRT

! ***************
! PROGRAM DOLGU ĞKAYNAI 2. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.
! ***************

CALL SEAMEND

! BILESENIN KALITESINI CAROLA PROGRAMINDA AYARLAYIN
! DAHA OENCE ARAYIN COMPEND
CALL COMPQMAN
! VEYA
! CALL COMPQAUT

! BILESENIN SONU
! COMPQMAN IÇIN ARAYIN VEYA COMPQAUT
CALL COMPEND

! BILESEN PROGRAMININ SONU
END

2.1.4 Karmaşık bileşen programı

Aşağıdaki Carola programı iki üretim adımına sahip bir bileşen oluşturur. Üretim adımları bir robot üzerinde sırayla veya
birkaç robot üzerinde paralel olarak kaynaklanabilir. Bunun için ön koşul, adımların bir protokolde birleştirilmesidir:

• Tip, varyant ve seri numarası tüm robotlarda aynı şekilde ayarlanmalıdır.

• İlgili bileşen şablonunda doğruminimumüretim adımı sayısı ayarlanmış olmalıdır. Bu durumda, iki üretim adımı kay-
naklanır, bu nedenle sayı en az iki değerine ayarlanmalıdır.

RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! ***************
! ÜRETIM ADIMI 1
! ***************

COMPTYPE := 'KAZAN'
COMPVARI := '80KW'
COMPSERI := '0042012342'
COMPSTEP := 'ZEMIN'

CALL COMPSTART

COMPSEAM := 'DOLGU ĞKAYNAI 1'
CALL SEAMSTRT
! PROGRAM DOLGU ĞKAYNAI 1. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.

COMPSEAM := 'DOLGU ĞKAYNAI 2'
CALL SEAMSTRT
! PROGRAM DOLGU ĞKAYNAI 2. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! ***************
! ÜRETIM ADIMI 2
! ***************

COMPTYPE := 'KAZAN'
COMPVARI := '80KW'
COMPSERI := '0042012342'
COMPSTEP := 'KAPAK'

CALL COMPSTART

COMPSEAM := 'DOLGU ĞKAYNAI 3'
CALL SEAMSTRT
! PROGRAM DOLGU ĞKAYNAI 3. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.

COMPSEAM := 'DOLGU ĞKAYNAI 4'

8



CALL SEAMSTRT
! PROGRAM DOLGU ĞKAYNAI 4. ĞÖRNEIN, ÇESITLI GC HAREKETLERI.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

3 Notlar

PDMveyaUMSmüşterisiyseniz ve kontrol ünitesinin konfigürasyonunda iki seçenektenbiri etkinleştirildiyse, aşağıdaki nota
acilen uyulmalıdır! İki seçenekten biri etkinleştirilirse, ancak robot kontrol ünitesi ilgili PC uygulamasına bağlı değilse, veri
akışında 5 saniyeye kadar uzayabilen ve bileşen protokolünü önemli ölçüde tahrif eden boşluklar olacaktır. Lütfen ya robot
kontrolörünü PC uygulamasına bağlayın ya da ilgili seçeneği yapılandırmadan kaldırın.

Aynı bileşen üzerinde aynı anda birden fazla robotla çalışıyorsanız ve tüm verilerin C-Gate.QMmodülündeki tek bir bileşen
protokolünde birleştirilmesini istiyorsanız, lütfen daha fazla bilgi için CLOOS Servisi ile iletişime geçin. Aksi takdirde, ge-
nel ekipman etkinliği (OEE) hesaplamasında istenmeyen yan etkilere yol açabileceğinden lütfen COMPSERI alanını boş
bırakın.

4 Değişkenler

Carola programından bileşen protokolüne değişkenler ekleme seçeneğiniz vardır. Bunu yapmak için OPCUAVAR komutunu
kullanın (örneğebakın). Eğer isim (sonparametre) ’COMPVAR_’ ön ekine sahipse, değişkenbileşene eklenir. Eğer isim ’STEP-
VAR_’ ön ekine sahipse, değişken bileşen adımına eklenir. İsim ’SEAMVAR_’ önekine sahipse, değişken geçerli dikişe eklenir.
Seçilen her ad için yalnızca bir değer saklanabilir. İlgili isimle en son ayarlanan değer günlükte saklanır. İsmin ön eki sak-
lanmaz. Dikiş, adım veya bileşen tamamlanmadan önce değer sıfırlanmalıdır. Bu yapılmazsa, değer bir sonraki öğeye de
eklenir. Tamsayılar, virgüllü sayılar ve metinler kaydedilebilir.

CALL SEAMSTRT
OPCUAVAR(0,1, 'EXAMPLE-VALUE','COMPVAR_SEAM_MEASURE_A')

OPCUAVAR(0,1, '','')
CALL SEAMEND

